


Gabarito – exercícios de apoio semana 1

### Capítulo 3 – Exercício 4a


Ler uma temperatura em graus Celsius e apresentá-la convertida em graus Fahrenheit. A fórmula de conversão é  $F \leftarrow C * 9 / 5 + 32$ , sendo F a temperatura em Fahrenheit e C a temperatura em Celsius.



```
programa TEMPERATURA
var
    C, F : real
início
    leia C
    F ← C * 9 / 5 + 32
    escreva F
fim
```

### Capítulo 4 – Exercício 3c


Realizar a leitura dos valores de quatro notas escolares bimestrais de um aluno representadas pelas variáveis N1, N2, N3 e N4. Calcular a média aritmética (variável MD) desse aluno e apresentar a mensagem "Aprovado" se a média obtida for maior ou igual a 5; caso contrário, apresentar a mensagem "Reprovado". Informar também, após a apresentação das mensagens, o valor da média obtida pelo aluno.



```
programa MÉDIA
var
    MD, N1, N2, N3, N4 : real
início
    leia N1, N2, N3, N4
    MD ← (N1 + N2 + N3 + N4) / 4
    se (MD ≥ 5) então
        escreva "Aluno aprovado com média: ", MD
    senão
        escreva "Aluno reprovado com média: ", MD
    fim_se
fim
```

## Capítulo 5 – Exercício 1c

Construir um programa que apresente a soma dos cem primeiros números naturais ( $1 + 2 + 3 + \dots + 98 + 99 + 100$ ).




```
programa SOMATÓRIO
var
    SOMA, CONTADOR : inteiro
início
    SOMA ← 0
    para CONTADOR de 1 até 100 passo 1 faça
        SOMA ← SOMA + CONTADOR
    fim_para
    escreva SOMA
fim
```

## Capítulo 6 – Exercício 1g

Elaborar um programa que leia duas matrizes do tipo vetor para o armazenamento de nomes de pessoas, sendo a matriz A com 20 elementos e a matriz B com 30 elementos. Construir uma matriz C, sendo esta a junção das matrizes A e B. Desta forma, a matriz C deve ter a capacidade de armazenar 50 elementos. Apresentar os elementos da matriz C.

```
programa JUNÇÃO
var
    A : conjunto[1..20] de inteiro
    B : conjunto[1..30] de inteiro
    C : conjunto[1..50] de inteiro
    I : inteiro
início
    para I de 1 até 20 passo 1 faça
        leia A[I]
    fim_para
    para I de 1 até 30 passo 1 faça
        leia B[I]
    fim_para
    para I de 1 até 50 passo 1 faça
        se (I <= 20) então
            C[I] ← A[I]
        senão
            C[I] ← B[I - 20]
        fim_se
    fim_para
    para I de 1 até 50 passo 1 faça
        escreva C[I]
    fim_para
fim
```



## Capítulo 7 – Exercício 1a

Elaborar um programa que leia 12 elementos numéricos inteiros em uma matriz do tipo vetor. Coloque-os em ordem decrescente e apresente os elementos ordenados.

```
programa CLASSIFICAÇÃO
var
  A : conjunto[1..12] de inteiro
  I, X, J : inteiro
início
  para I de 1 até 12 passo 1 faça
    leia A[I]
  fim_para
  para I de 1 até 11 passo 1 faça
    para J de I + 1 até 12 passo 1 faça
      se (A[I] < A[J]) então
        X ← A[I]
        A[I] ← A[J]
        A[J] ← X
      fim_se
    fim_para
  fim_para
  para I de 1 até 12 passo 1 faça
    escreva A[I]
  fim_para
fim
```

